Comparing the derivation of equation of motion for double pendulum by method of F=ma and by energy (Lagrangian) method

by Nasser M. Abbasi

May 11, 2011

The goal of this note is to show how to use a symbolic program to help solve a typical basic engineering problem that requires large amount of algebraic manipulation. By applying a CAS program such as *Mathematica* on such a problem, one can then concentrate more on the problem formulation itself by having the program do the long algebraic computation, which can be very time consuming if done by hand, and also can be error prone.

This also allows one to be able to experiment with different formulation, as one is now relieved from the time consuming part and is able to try different solutions and different idea, which can lead to better insight of the problem.

In this note, we solve the double pendulum problem both ways. One way is by using the energy method, and another way is by using F=ma approach. Both would require long time to do by hand due to the large amount of algebra involved. And at the end, both solutions are verified to be the same.

download PDF NOTEBOOK

system diagram

This diagram shows the velocity and acceleration components of the 2 masses in the double pendulum.



Lagrangian solution

Write down the kinetic energy T

In[17]:=
```
T = 1/2 m₁ (L₁ θ'[t])^2 + 1/2 m₂ ((L₂ (ϕ'[t]) + L₁ θ'[t] Cos[ϕ[t] - θ[t]])^2 +
      (L₁ θ'[t] Sin[ϕ[t] - θ[t]])^2);
```

Out[18]=

$$T=\frac{1}{2} L_1^2 m_1 \dot{\theta}^2 + \frac{1}{2} m_2 \left(L_1^2 \dot{\theta}^2 + 2 L_1 L_2 \operatorname{Cos}[\theta - \phi] \dot{\theta} \dot{\phi} + L_2^2 \dot{\phi}^2\right)$$

Write down the potential energy

In[19]:=
```
V = -L₁ m₁ g Cos[θ[t]] - (L₁ Cos[θ[t]] + L₂ Cos[ϕ[t]]) m₂ g;
```

Out[20]=

$$V = -g L_1 m_1 \mathrm{Cos}[\theta] - g m_2 (L_1 \mathrm{Cos}[\theta] + L_2 \mathrm{Cos}[\phi])$$

Write down the Lagrangian L

In[21]:=

```
L = T - V;
```

Out[22]=

$$L = \frac{1}{2}\left(2 g (L_1 (m_1 + m_2) \mathrm{Cos}[\theta] + L_2 m_2 \mathrm{Cos}[\phi]) + L_1^2 (m_1 + m_2) \dot{\theta}^2 + 2 L_1 L_2 m_2 \mathrm{Cos}[\theta - \phi] \dot{\theta} \dot{\phi} + L_2^2 m_2 \dot{\phi}^2\right)$$

Find equation of motion for θ

In[26]:=

```
eq1 = D[D[L, θ'[t]], t] - D[L, θ[t]];
sol1 = First@Solve[eq1 == 0, θ''[t]];
```

Out[28]=

$$\ddot{\theta} = -\frac{L_2 m_2 \mathrm{Cos}[\theta - \phi] \ddot{\phi} + g (m_1 + m_2) \mathrm{Sin}[\theta] + L_2 m_2 \dot{\phi}^2 \mathrm{Sin}[\theta - \phi]}{L_1 (m_1 + m_2)}$$

Find equation of motion for ϕ

In[29]:=

```
eq2 = D[D[L, ϕ'[t]], t] - D[L, ϕ[t]];
sol2 = First@Solve[eq2 == 0, ϕ''[t]];
```

Out[31]=

$$\ddot{\phi} = -\frac{L_1 \mathrm{Cos}[\theta - \phi] \ddot{\theta} - L_1 \dot{\theta}^2 \mathrm{Sin}[\theta - \phi] + g \mathrm{Sin}[\phi]}{L_2}$$

decouple the above 2 equations

In[32]:=

```
sol = First@Solve[{eq1 == 0, eq2 == 0}, {θ''[t], ϕ''[t]}];
```

Out[33]=

$$\ddot{\theta} = -\frac{g (2 m_1 + m_2) \mathrm{Sin}[\theta] + g m_2 \mathrm{Sin}[\theta - 2 \phi] + 2 m_2 \left(L_1 \mathrm{Cos}[\theta - \phi] \dot{\theta}^2 + L_2 \dot{\phi}^2\right) \mathrm{Sin}[\theta - \phi]}{2 L_1 \left(m_1 + m_2 \mathrm{Sin}[\theta - \phi]^2\right)}$$

Out[34]=

$$\ddot{\phi} = \frac{\left((m_1 + m_2)\left(g\,\mathrm{Cos}[\theta] + L_1\,\dot{\theta}^2\right) + L_2\,m_2\,\mathrm{Cos}[\theta-\phi]\,\dot{\phi}^2\right)\mathrm{Sin}[\theta-\phi]}{L_2\left(m_1 + m_2\,\mathrm{Sin}[\theta-\phi]^2\right)}$$

Numerically solve the 2 equations found above for θ(t) and ϕ(t), use some arbitrary parameters

In[38]:=

```
pars = {L1 -> 1, L2 → 1, m1 → 1, m2 → 1, g → 9.8};
ic = {θ[0] == 1.2 , θ'[0] == 0, ϕ[0] == 3, ϕ'[0] == 0};
numericalSolution = First@
  NDSolve[Flatten[{eq1 == 0, eq2 == 0, ic}] /. pars, {θ[t], ϕ[t]}, {t, 0, 10}]
```

Out[40]=

```
{θ[t] → InterpolatingFunction[{{0., 10.}}, <>][t],
 ϕ[t] → InterpolatingFunction[{{0., 10.}}, <>][t]}
```

Make plots of the solutions. Assume
$L_1 = 1$, $L_2 = 1$, $m_1 = 1$, $m_2 = 1$, $\theta(0) = 1$, $\theta'(0) = 0$, $\phi(0) = 3$, $\phi'(0) = 0$

In[41]:=

```
frameLabel[t_] := {{None, None}, {"t", t}}
at = {Frame → True, ImageSize → 250};

Row[{Plot[θ[t] /. numericalSolution, {t, 0, 10}, Evaluate@at,
   FrameLabel -> frameLabel["θ(t)"]], Plot[ϕ[t] /. numericalSolution,
   {t, 0, 10}, Evaluate@at, FrameLabel -> frameLabel["ϕ(t)"]]}]
```

Out[43]=



F=ma solution

Write the free body diagram



We have 4 unknowns T_1, T_2, $\theta(t)$, $\phi(t)$

Write F=ma for forces radially along θ along for m_1

```
eq1 = T1 - m1 g Cos[θ[t]] - T2 Cos[ϕ[t] - θ[t]] == m1 (L1 θ'[t]^2);
```

$$eq1 = T1 == g\, m_1 \operatorname{Cos}[\theta] + T2 \operatorname{Cos}[\theta - \phi] + L_1\, m_1\, \dot{\theta}^2$$

Resolve forces perpendicular for m_1

```
eq2 = T2 Sin[ϕ[t] - θ[t]] - m1 g Sin[θ[t]] == m1 (L1 θ''[t]);
```

$$eq2 = L_1\, m_1\, \ddot{\theta} + g\, m_1 \operatorname{Sin}[\theta] + T2 \operatorname{Sin}[\theta - \phi] == 0$$

Write F=ma for forces radially along ϕ along for m_2

```
eq3 = T2 - m2 g Cos[ϕ[t]] ==
    m2 (L1 θ'[t]^2 Cos[ϕ[t] - θ[t]] + L2 ϕ'[t]^2 - L1 θ''[t] Sin[ϕ[t] - θ[t]]);
```

$$eq3 = T2 == m_2 \left(g \operatorname{Cos}[\phi] + L_1 \operatorname{Cos}[\theta - \phi]\, \dot{\theta}^2 + L_2\, \dot{\phi}^2 + L_1\, \ddot{\theta} \operatorname{Sin}[\theta - \phi]\right)$$

Resolve forces perpendicular for m_2

```
eq4 = -m1 g Sin[ϕ[t]] ==
    m1 (L2 ϕ''[t] + L1 θ''[t] Cos[ϕ[t] - θ[t]] + L1 θ'[t]^2 Sin[ϕ[t] - θ[t]]);
```

$$\text{eq4} = m_1 \left(L_1 \operatorname{Cos}[\theta-\phi]\, \ddot{\theta} + L_2\, \ddot{\phi} - L_1\, \dot{\theta}^2 \operatorname{Sin}[\theta-\phi] + g \operatorname{Sin}[\phi]\right) == 0$$

Solve the above 4 equestions

```
solF = First@Solve[{eq1, eq2, eq3, eq4}, {T1, T2, θ''[t], ϕ''[t]}];
```

$$T_1 = \frac{m_1 \left((m_1+m_2)\left(g \operatorname{Cos}[\theta] + L_1 \dot{\theta}^2\right) + L_2 m_2 \operatorname{Cos}[\theta-\phi]\, \dot{\phi}^2\right)}{m_1 + m_2 \operatorname{Sin}[\theta-\phi]^2}$$

$$T_2 = \frac{2 m_1 m_2 \left(\operatorname{Cos}[\theta-\phi]\left(g \operatorname{Cos}[\theta] + L_1 \dot{\theta}^2\right) + L_2 \dot{\phi}^2\right)}{2 m_1 + m_2 - m_2 \operatorname{Cos}[2(\theta-\phi)]}$$

$$\ddot{\theta} = -\frac{m_2 \left(L_1 \operatorname{Cos}[\theta-\phi]\, \dot{\theta}^2 + L_2 \dot{\phi}^2\right) \operatorname{Sin}[\theta-\phi] + g\left(m_1 \operatorname{Sin}[\theta] + m_2 \operatorname{Cos}[\phi] \operatorname{Sin}[\theta-\phi]\right)}{L_1 \left(m_1 + m_2 \operatorname{Sin}[\theta-\phi]^2\right)}$$

$$\ddot{\phi} = \frac{\left((m_1+m_2)\left(g \operatorname{Cos}[\theta] + L_1 \dot{\theta}^2\right) + L_2 m_2 \operatorname{Cos}[\theta-\phi]\, \dot{\phi}^2\right) \operatorname{Sin}[\theta-\phi]}{L_2 \left(m_1 + m_2 \operatorname{Sin}[\theta-\phi]^2\right)}$$

We see that the solution from F=ma is the same as the solution obtained from Lagrange method. To verify

```
(θ''[t] /. sol) - (θ''[t] /. solF) // Simplify
```

```
0
```

```
(ϕ''[t] /. sol) - (ϕ''[t] /. solF) // Simplify
```

```
0
```

We see from the above that the solutions are identical.